

Werner-von-Siemens-Gymnasium Weißenburg/Bay.

Kollegstufe

Abiturjahrgang 1991

Facharbeit aus der Physik

THEMA: Der Tanz der Trojaner

Verfasser: Andreas Würfl

Leistungskurs: Ph₃₂

Kursleiter: Herr Bauer

Bearbeitungszeitraum: 15.2.1990 – 18.1.1991

Abgabetermin: 18.1.1991

Vom Kursleiter auszufüllen:

Erzielte Note:

Erzielte Punkte:
(einfache Wertung)

Eintragung der Ergebnisse:
(Datum)

Dem Direktorat vorgelegt:
(Datum)

.....
(Kursleiter)

Inhaltsverzeichnis

| | | |
|----------|---|-----------|
| 1 | Einführung | 3 |
| 1.1 | Kleine Planeten (Asteroiden, Planetoiden) | 3 |
| 1.2 | Kleinplaneten und Himmelsmechanik | 5 |
| 2 | Die Trojaner | 6 |
| 2.1 | Das Dreikörperproblem von Lagrange | 7 |
| 2.2 | Die Berechnung der Gleichgewichtslösungen . . . | 8 |
| 3 | Berechnung eines anderen Lagrangepunktes | 9 |
| 3.1 | Dimensionsloses Rechnen | 9 |
| 3.2 | Die Umsetzung in ein PASCAL-Programm | 10 |
| 3.2.1 | Die Platzierung des Satelliten | 11 |
| 3.2.2 | Die Kräfte auf den Satelliten | 13 |
| 3.2.3 | Das Problem der Differentialgleichungen . | 15 |
| 3.2.4 | Die Resultate der Berechnung | 15 |
| | Literatur | 19 |
| | Programmlisting | 20 |
| 4 | Erklärung | 27 |

1 Einführung

1.1 Kleine Planeten (Asteroiden, Planetoiden)

Bereits sehr früh fiel es den Astronomen auf, daß die Abstände der einzelnen Planeten von der Sonne einer gewissen Gesetzmäßigkeit unterliegen. Je weiter sie von der Sonne, dem Zentralgestirn unseres Planetensystems, entfernt sind, desto größer werden die Lücken zwischen den Planetenbahnen. Christian Wolf wies im Jahre 1741 als erster auf diese Merkwürdigkeit hin, für die später, im Jahre 1772, von den Astronomen Johann Daniel Titius und Johann Elert Bode ein mathematisch strenger gefaßtes Gesetz gefunden wurde:

$$a = \frac{1}{10}(4 + 3 \cdot 2^n)$$

Für alle damals bekannten Planeten lieferte diese Gleichung für $n = -\infty, 0, 1, 2, 4, 5$ näherungsweise die mittlere Entfernung zur Sonne in AE¹. Der im Jahre 1781 entdeckte Planet Uranus stellte einen weiteren Beweis für die Gültigkeit dieses Gesetzes dar. Nur zwischen Mars und Jupiter, in einem Sonnenabstand von ungefähr 2,8 AE, war eine Lücke, obwohl sich dort nach der Titius-Bode-Regel für den Wert $n = 3$ ein Planet befinden müßte. Die Astronomen rätselten, ob an dieser Stelle nicht doch noch ein weiterer, bisher nicht entdeckter Planet um die Sonne kreise. Bis dahin hatte aber noch nichts auf die Existenz eines solchen Planeten hingewiesen, der aber vielleicht wegen seiner Lichtschwäche den Beobachtern früher entgangen sein könnte.

Um die Existenz des unbekannten Planeten, die bisher nur auf mathematischen Überlegungen beruhte, zu beweisen, schlossen sich Ende des 18. Jahrhunderts einige Astronomen und Sternwarten zusammen. Man versuchte den bisher nicht entdeckten Planeten zwischen Mars und Jupiter im Vertrauen auf

¹ AE = astronomische Einheit; $1AE = 1,495978 \cdot 10^{11}m$ (Länge der großen Halbachse der Erdbahn)

die Richtigkeit des Titius-Bodeschen Gesetzes zu finden. Der erste Astronom, der Erfolge bei der Suche hatte, war der Mönch Guiseppe Piazzi, der in Palermo gerade damit beschäftigt war, einen Sternkatalog herzustellen. Piazzi wollte sich von der Richtigkeit seiner Angaben überzeugen, und so fand er bei seinen Sternbeobachtungen in der Neujahrsnacht 1800/1801 zu seiner großen Verwunderung ein Sternchen, dessen Helligkeit nur achte Größe war, das sich im Laufe seiner Beobachtungen weiterbewegte; das konnte kein Fixstern sein! Piazzi hatte einen neuen Planet entdeckt, aber er war sich noch nicht sicher, in welchem Abstand zur Sonne sich dieser Himmelskörper bewegte. Leider verschwand der Planet am 11. Februar im Strahlenbereich der Sonne, und für Piazzi reichten die wenigen Daten des Sternchens, die aus seinen Beobachtungen hervorgingen, kaum aus, mit den damaligen Mitteln der Mathematik den weiteren Verlauf der Bahn des neuentdeckten Planeten zu berechnen und ihn somit wiederzufinden.

Piazzi veröffentlichte seine Beobachtungen, und der vierundzwanzigjährige Mathematiker Carl Friedrich Gauß erfuhr so von dessen Problem. Nachdem der französische Mathematiker Laplace eine Berechnung der Bahn des neuentdeckten Planeten für unmöglich erachtet hatte, gelang es Gauß, eine neue Methode der Bahnbestimmung zu entwickeln, nach der bis auf den heutigen Tag die Bahnen neuentdeckter Kometen oder Planeten berechnet werden. Nach seinen Vorausberechnungen konnte der Gothaer Astronom Zach am 7. Dezember 1801 den nun „Ceres“ genannten Planeten tatsächlich wieder auffinden.

Nach und nach wurden immer mehr solcher Kleinkörper, die sogenannten *kleinen Planeten*, *Asteroiden* oder *Planetoiden* entdeckt. Die Schätzungen über die Gesamtmasse der *Planetoiden* reichen von 10^{21} kg bis 10^{22} kg. Bis zum Ende des 19. Jahrhunderts wurden weitere hundert solcher *Asteriode* gefunden. Erst

als durch Max Wolf in Heidelberg 1890 erstmals die Himmelsphotographie zur Suche nach diesen, sich bewegenden Himmelskörpern eingesetzt wurde, wuchs ihre Anzahl schnell. Zur Jahreswende 1962/63 waren schon 1651 Planetoiden gefunden und katalogisiert.

1.2 Kleinplaneten und Himmelsmechanik

Die Störungstheorie der Himmelsmechanik fordert, daß die Umlaufzeiten zweier Planeten nicht kommensurabel, d.h. nicht im Verhältnis kleiner ganzer Zahlen zueinander stehen dürfen. Dieses wirkt sich auch auf die Planetoiden aus, denn im dichten Schwarm der Planetoiden gibt es seltsamerweise Lücken. Diese Lücken entsprechen den Umlaufzeiten, die in einem ganzzahligen Verhältnis zur Umlaufzeit des Jupiter stehen. Die Gravitationsstörungen des Planeten Jupiter wirken so auf die Kleinplaneten ein, daß diese aus den Bereichen, in denen Kommensurabilität auftritt, herausgedrängt wurden. So hat der größte und massereichste aller Planeten des Sonnensystems, der Jupiter, die Schar der Planetoiden so geordnet, wie wir sie heute vorfinden.

Seltsamerweise halten sich einige der Planetoiden aber nicht an die Störungstheorie der Himmelsmechanik, ihre Umlaufzeiten stehen tatsächlich im Verhältnis kleiner ganzer Zahlen zu der Umlaufzeit des Jupiter. Es muß also doch möglich sein, die Störungstheorie der Himmelsmechanik zu umgehen. Man hat diese Planetoiden nach ihrer Resonanz² in verschiedene Gruppen eingeteilt:

- Die Trojaner (22 Stck., Resonanz $\frac{1}{1}$), siehe Tab. 1
- Die Hilda-Gruppe (27 Stck., Resonanz $\frac{2}{3}$)
- Die „Ungarischen“ (16 Stck., Resonanz $\frac{1}{4}$)

²Die Resonanz entspricht dem Verhältnis der Umlaufzeiten zweier betrachteter Körper.

| Nr. | Name | mag | i | e | μ | r | ϱ |
|------|------------|------|------|------|-------|------|-----------|
| 588 | Achilles | 16.0 | 10.3 | 0.15 | 298 | +44° | 55 km |
| 617 | Patroclus | 15.8 | 22.1 | 0.14 | 299 | −63 | 60 |
| 624 | Hector | 15.2 | 18.3 | 0.02 | 306 | +70 | 78 |
| 659 | Nestor | 16.3 | 4.5 | 0.11 | 296 | +74 | 48 |
| 884 | Priamus | 16.5 | 8.9 | 0.12 | 298 | −82 | 45 |
| 911 | Agamemnon | 15.4 | 21.9 | 0.07 | 305 | +69 | 70 |
| 1143 | Odysseus | 16.0 | 3.1 | 0.09 | 300 | +66 | 55 |
| 1172 | Aeneas | 16.0 | 16.7 | 0.10 | 300 | −76 | 55 |
| 1173 | Anchises | 16.6 | 7.0 | 0.14 | 308 | −45 | 40 |
| 1208 | Troilus | 16.3 | 33.7 | 0.09 | 303 | −60 | 47 |
| 1404 | Ajax | 16.8 | 18.1 | 0.11 | 302 | +85 | 37 |
| 1437 | Diomedes | 15.8 | 20.6 | 0.04 | 304 | +45 | 61 |
| 1583 | Antilochus | 16.5 | 28.3 | 0.05 | 293 | +33 | 47 |
| 1647 | Menelaus | 18.5 | 5.6 | 0.03 | 299 | +71 | 18 |

In der Tabelle ist in Spalte 3 (mag) die photographische Helligkeit zur mittleren Opposition gegeben; unter i die Neigung der Bahn gegen die Ekliptik; unter e die Bahnexzentrizität; die mittlere tägliche Bewegung μ ist in Bogensekunden gegeben, für Jupiter beträgt sie 299''; r ist die Distanz von Jupiter in Grad der Länge für die Epoche 1960 Juni 20; ϱ , der Durchmesser des Planetoiden in Kilometern, wurde aufgrund seiner Helligkeit gemessen. Quelle: [5]

Tabelle 1: 14 der 22 bekannten Trojaner

- Die Apollo–Amor–Objekte (48 Stck.)

Das Phänomen der Trojaner, die sich mit der gleichen Geschwindigkeit und dem gleichen Radius wie der Jupiter um die Sonne bewegen, soll im folgenden genauer untersucht werden.

2 Die Trojaner

Die Planetoiden der Trojanergruppe (siehe Tabelle 1) kreisen zusammen mit dem Gasriesen Jupiter um die Sonne; ein Teil davon, die *Achilles-Gruppe*, eilt auf der Jupiterbahn um ca. 60° voraus, der andere Teil, die *Patroclus-Gruppe*, folgt Jupiter auf seiner Bahn mit einem Winkelabstand von ca. 60°. Dabei beschreiben sie noch Eigenbewegungen in nierenförmigen Bahnen.

2.1 Das Dreikörperproblem von Lagrange

Schon lange vor der Entdeckung der Kleinplaneten machte sich der französische Mathematiker Joseph Luis Lagrange Gedanken über die Wechselwirkungen dreier Körper im Raum; er beschrieb einen Sonderfall des Dreikörperproblems, bei dem zwei Körper durch ihre Massen eine merkliche Schwerkraft ausüben und ein dritter fast masseloser Körper im Schwerfeld der anderen Körper treibt. Er bewies die Existenz kräftefreier Punkte, sog. Librationspunkte, an denen sich die auf den fast masselosen Körper wirkenden Gravitations- und Fliehkräfte aufheben. An solchen Librationspunkten können also Körper kleiner Masse für längere Zeit kräftefrei verweilen. Der geometrische Ort dieser Librationspunkte in einem System, das aus zwei um ihren gemeinsamen Schwerpunkt S rotierenden Körpern M_1 und M_2 besteht, ist im wesentlichen konstant. Drei Librationspunkte liegen auf der die zwei Körper verbindenden Geraden (L_1, L_2, L_3 in Bild 1) zwei weitere (L_4 und L_5) bilden mit den Körpern gleichschenklige Dreiecke. Wird ein Massepunkt in einem der ersten drei Punkte mit der richtigen Geschwindigkeit positioniert, so befindet er sich anfänglich im Gleichgewichtszustand. Allerdings wird er, wenn er sich nur ein wenig vom Lagrangepunkt entfernt, von den auftretenden Kräften noch weiter weggezogen! Die übrigen zwei Librationspunkte L_4 und L_5 bilden mit den beiden Massen M_1 und M_2 ein gleichseitiges Dreieck. An diesen Punkten wird der Körper von den auftretenden Kräften festgehalten, wenn er sich etwas von seinem Lagrangeschen Punkt entfernt. Es kommt dann zu Schwingungen um diese Punkte.

Das Verhalten der Trojaner läßt sich mit Hilfe dieser Theorie erklären. Die beiden massereichen Körper, die um ihren gemeinsamen Schwerpunkt kreisen, sind im Falle der Trojaner Sonne und Jupiter. Die Trojaner, im Vergleich dazu fast masselos, schwingen um die Librationspunkte L_4 und L_5 , die mit




Abbildung 1: Die Lage der fünf Librationspunkte

Sonne und Jupiter jeweils ein gleichseitiges Dreieck bilden. Die Schwingung kommt dadurch zustande, daß die Körper den Ort der absoluten Kräftelosigkeit nie erreichen und sich wie ein Pendel um den Punkt der idealen Kräftelosigkeit bewegen, ohne längere Zeit dort zu verweilen.

2.2 Die Berechnung der Gleichgewichtslösungen

Zur Lösung des Problems betrachtet man die Bewegung eines Massepunktes im Gravitationsfeld zweier Körper großer Masse (M_1, M_2), mit dem Massenverhältnis $c = \frac{M_1}{M_2}$, z.B. Sonne und Jupiter im Abstand r , die ihren gemeinsamen Schwerpunkt S auf elliptischen Bahnen umlaufen. Berechnet man nun die auf den punktförmigen Körper wirkenden Gravitations- und Fliehkräfte, so kann man, für den zweidimensionalen Fall, nach der Lagrange-Methode zwei Differentialgleichungen zweiter Ordnung aufstellen, als deren Gleichgewichtslösungen man die Librationspunkte findet. Wenn man nun noch den Ort des punktförmigen, masselosen Körpers in korotierenden Koordinaten berechnet und aufzeichnet, kann man die Bahn des Körpers im Kräftefeld der beiden massereichen Körper bestim-




Abbildung 2: Eine Skizze der Vereinfachungen

men. Physikalisch gesehen sind diese Lösungen für den Fall von kreisförmigen Bahnen stationär, während sich für elliptische Bahnen, die Abstände der Librationspunkte von M_1 und M_2 periodisch ändern.

3 Berechnung eines anderen Lagrangepunktes

Zur Veranschaulichung des Effektes von Lagrange werden wir nun ein einfaches Beispiel verwenden: Wir betrachten ein System Erde–Mond–Satellit, wobei Erde und Mond die beiden Massen M_1 und M_2 darstellen, die Masse des Satelliten ist im Verhältnis zu Erde und Mond vernachlässigbar klein. Wir wollen versuchen den Satelliten an den Lagrangepunkt L_2 zu setzen und seine weitere Flugbahn berechnen.

3.1 Dimensionsloses Rechnen

Um den Aufwand bei der Umsetzung in ein Pascalprogramm möglichst gering zu halten und um einer besseren Übersicht willen, versuchen wir eine Verwendung von nicht ganzzahligen Konstanten und Größen, wie z.B. die Gravitationskonstante G und den genauen Massewerten von Erde und Mond zu vermeiden.

Die Vereinfachungen sehen folgendermaßen aus:

1. **Die Entfernungen:** Es wird festgesetzt, daß die Entfernung Erde–Mond die Länge ,1‘ hat. Die Entfernung Erde - gemeinsamer Schwerpunkt bekommt die Bezeichnung c und die Entfernung Mond - gemeinsamer Schwerpunkt ist somit $1 - c$ (siehe Abb. 2).
2. **Die Massen:** Wenn man die Zentripetalkräfte für Erde und Mond berechnet und die beiden entstehenden Gleichungen dann dividiert bekommt man eine Beziehung zwischen den Massen der Erde und des Mondes.

$$\begin{aligned}
 F_{Z,Erde} &= m_{Erde} \cdot \omega^2 \cdot c; \\
 F_{Z,Mond} &= m_{Mond} \cdot \omega^2 \cdot (1 - c); \\
 \frac{F_{Z,Erde}}{F_{Z,Mond}} &\mapsto \frac{m_{Erde}}{m_{Mond}} \cdot \frac{c}{1 - c} = 1 \implies \frac{m_{Erde}}{m_{Mond}} = \frac{1 - c}{c};
 \end{aligned}$$

Zur weiteren Vereinfachung kann man nun für die Masse der Erde und des Mondes folgendes einsetzen: $m_{Erde} = 1 - c$ und $m_{Mond} = c$.

3. **Die Winkelgeschwindigkeit:** Wenn wir die bisher vereinfachten Werte zur Hilfe nehmen, können wir die Zentripetalkraft F_Z und die Gravitationskraft F_G gleichsetzen.

$$\begin{aligned}
 F_{Z,Erde} &= F_{Z,Mond} = F_G; \\
 (1 - c) \cdot \omega^2 \cdot c &= (1 - c) \cdot c \implies \omega^2 = 1; \\
 \omega &= 1;
 \end{aligned}$$

Somit hat die Winkelgeschwindigkeit ω den vereinfachten Wert ,1‘. Dafür gelten folgende Bedingungen: $G=1$, $m_{Erde} = 1 - c$, $m_{Mond} = c$ und $c \ll 1$.

3.2 Die Umsetzung in ein PASCAL-Programm

Nachdem die Berechnung auf dem Computer durch das dimensionslose Rechnen sehr vereinfacht wurde, kann man nun dazu

übergehen, die Berechnung der Bahn des Satelliten im Gravitationsfeld von Erde und Mond schrittweise zu entwickeln.

3.2.1 Die Platzierung des Satelliten

Zu Beginn des Programmlaufes muß der Satellit möglichst in die Nähe eines Lagrangeschen Punktes gebracht werden, damit er sich von Anfang an in der Gleichgewichtslage befindet. Zu diesem Zweck wird der Punkt berechnet, an dem sich alle auf den Satelliten einwirkenden Kräfte, wie Flieh- und Gravitationskräfte, aufheben. Die Kräfte, die auf den Satelliten einwirken sind:

- die Gravitation der Erde $F_{G,Erde}$
- die Gravitation des Mondes $F_{G,Mond}$
- die Zentripetalkraft F_Z

Wobei die beiden letztgenannten Kräfte gemeinsam entgegengesetzt zur Gravitationskraft der Erde wirken. Wenn man alle Kräfte in einer Gleichung betrachtet, kann man den Ort berechnen, an dem sich diese gerade aufheben:

$$\begin{aligned} F_{G,Erde} &= F_{G,Mond} + F_Z \\ \frac{m_S \cdot m_E}{(r_S + c)^2} &= \frac{m_S \cdot m_M}{(1 - r_S - c)^2} + m_S \cdot \omega^2 \cdot r_S \end{aligned}$$

Jetzt kann man die Masse des Satelliten vernachlässigen und die Winkelgeschwindigkeit $\omega = 1$ setzen und eine Funktion der auf den Satelliten in Abhängigkeit seines Ortes einwirkenden Kräfte aufstellen:

$$f(r_S) = \frac{1 - c}{(c + r_S)^2} - \frac{c}{(1 - c - r_S)^2} - r_S$$

Diese Funktion ist in Abb. 3 dargestellt. Die Nullstelle der Funktion ergibt sich mit Hilfe der Iterationsmethode von Newton. Nun platziert man den Satelliten in dieser Nullstelle und setzt




Abbildung 3: Die Funktion der Kräfte

seine Geschwindigkeit in y -Richtung auch auf den Betrag des x -Wertes der Nullstelle.

In Formeln sieht dies folgendermaßen aus:

$$\text{Positionsvektor: } \vec{x}_{\text{Satellit}} = \begin{pmatrix} x \\ 0 \end{pmatrix}$$

$$\text{Geschwindigkeitsvektor: } \vec{v}_{\text{Satellit}} = \begin{pmatrix} 0 \\ -x \end{pmatrix}$$

Das Pascalprogramm zur Newtoniteration:

```
x := 0;

repeat           {Newtonsches Iterationsverfahren}
  xa := x;
  m := f1(x);
  y := f0(x);
  t := y - m*x;
  x := -t/m;
```

```
until abs(f0(x))<0.001;
```

Erklärung: $f0$ stellt die zu untersuchende Funktion dar und $f1$ deren Ableitung. Das Programm beginnt an der Stelle $x = 0$ mit den Iterationen. Der Steigungsfaktor m und der Verschiebungsfaktor t der Tangente an dem untersuchten Punkt werden berechnet. Daraus ergibt sich die Nullstelle der Tangentengeraden, die nun als Ansatzpunkt für eine weitere Iteration dient. usw.. Wenn sich der Funktionswert nur noch kaum von 0 unterscheidet wird die Iterationsschleife abgebrochen. Die Variable x enthält nun den x -Wert der Nullstelle der Kräftefunktion. Der Satellit wird nun so positioniert, daß er auf der Verbindungsgeraden von Erde und Mond liegt und vom Schwerpunkt den soeben ausgerechneten Abstand x besitzt.

3.2.2 Die Kräfte auf den Satelliten

Damit man eine Aussage über die Bewegungen des Satelliten machen kann, muß man den resultierenden Beschleunigungsvektor der auf den Satelliten wirkenden Gravitationsbeschleunigungen der Erde und des Mondes berechnen. Das Gravitationsgesetz von Newton für die Gravitationskraft, mit der die Erde den Satelliten anzieht, lautet:

$$F_G = G \cdot \frac{m_E \cdot m_S}{r_{ES}^2}$$

Da man in diesem Falle für die Gravitationskonstante G den Wert 1 einsetzt, kann man G für die weiteren Berechnungen außer acht lassen. Aus der Gravitationskraft kann man dann die Gravitationsbeschleunigung a_G berechnen:

$$a_G = \frac{F_G}{m_S} = \frac{m_E}{r_{ES}^2}$$

Den Vektor der Gravitationsbeschleunigung bekommt man, wenn man noch den Richtungsvektor der Beschleunigung zum

Betrag der Gravitationsbeschleunigung dazumultipliziert:

$$\vec{a}_G = \frac{m_E}{r_{ES}^2} \cdot \frac{\vec{r}_{ES}}{|\vec{r}_{ES}|} = \frac{m_E}{r_{ES}^3} \cdot \vec{r}_{ES}$$

Der Vektor wird in seine x - und y -Komponente aufgespalten:

$$\vec{a}_G = \frac{m_E}{r_{ES}^3} \cdot \begin{pmatrix} x_S - x_E \\ y_S - y_E \end{pmatrix}$$

Wenn man die gleichen Schritte für die Berechnung der Gravitationsbeschleunigung des Satelliten durch den Mond durchläuft und nun die beiden Beschleunigungen zur Resultierenden addiert, kommt man zu folgendem Ergebnis:

$$\vec{a}_{gesamt} = \frac{m_E}{r_{ES}^3} \cdot \begin{pmatrix} x_S - x_E \\ y_S - y_E \end{pmatrix} + \frac{m_M}{r_{MS}^3} \cdot \begin{pmatrix} x_S - x_M \\ y_S - y_M \end{pmatrix}$$

Umgesetzt in PASCAL lautet diese Formel:

```

rM.x := Mond[Nr].x - Satellit.x;
rM.y := Mond[Nr].y - Satellit.y;
rE.x := Erde[Nr].x - Satellit.x;
rE.y := Erde[Nr].y - Satellit.y;

rM2 := sqr(rM.x) + sqr(rM.y);
rE2 := sqr(rE.x) + sqr(rE.y);
rM3 := rM2 + sqrt(rM2);
rE3 := rE2 + sqrt(rE2);

a.x := mE * rE.x / rE3 + mM * rM.x / rM3;
a.y := mE * rE.y / rE3 + mM * rM.y / rM3;
```

Erklärung: rM und rE sind die jeweiligen Abstände des Satelliten zum Mond bzw. zur Erde. rM3 und rE3 sind dann die Abstände hoch drei. Der Variablen a wird die x - und die y -Komponente der auf den Satelliten wirkenden Beschleunigung zugewiesen.

3.2.3 Das Problem der Differentialgleichungen

Um aus der oben hergeleiteten Berechnung der Beschleunigung und ihrer Richtung nun den neuen Ort und die neue Geschwindigkeit des Körpers zu errechnen, muß man je Koordinatenachse eine Differentialgleichung zweiter Ordnung zu Hilfe nehmen.


$$\begin{aligned}a_x &= x'' = f(t, x) \\a_y &= y'' = f(t, y) \quad .\end{aligned}$$

Wir stützen uns hier auf die Tatsache, daß die Beschleunigung a die zweite Ableitung des Ortes s darstellt.


Auf direktem Wege ist die Berechnung des neuen Ortes nicht möglich, da es für die Differentialgleichung keine Lösungsformel gibt, man muß sich deshalb einer geeigneten Näherungsformel für die Gleichungen bedienen. Zur näherungsweisen Berechnung der beiden Differentialgleichungen zweiter Ordnung wird das Verfahren von RUNGE-KUTTA-NYSTRÖM herangezogen, mit ihm läßt sich ohne großen Aufwand ein gutes Ergebnis erzielen, da es durch günstige Mittelwertbildung die Vorteile einer kurzen Ausführungszeit mit einer ausreichenden Genauigkeit verbindet.

3.2.4 Die Resultate der Berechnung

Für den Wert $c = 0.2$ und einer Berechnung von 200 Schritten ergibt sich folgendes Bild:




Der Satellit, der anfänglich lange an der stabilen Position des Librationspunktes L_2 verharrte, beschreibt schließlich doch eine Bahn um die Erde, da er durch die Kräfte, die bei einem geringen Abweichen vom Librationspunkt entstehen, weggezogen wurde. Das nächste Bild zeigt die Satellitenbahn für $c = 0.1$ mit Berechnung von 300 Schritten.



Auch hier zeigt sich, daß die Position des Satelliten im Librationspunkt L_2 sehr instabil ist. Auffällig ist jedoch, daß der Satellit mit zunehmender Erdmasse länger in seiner anfänglichen Position im Punkt L_2 verharrt. Auch die Bahn, die der Satellit um die Erde beschreibt, weicht mit zunehmender Erdmasse


von einer Kreisbahn ab. Dies zeigt sich besonders deutlich im nächsten Bild, das mit $c = 0.01$ auf 600 Schritte berechnet wurde:



Das Programm ist in der Lage, auch das Verhalten eines Körpers an den anderen Librationspunkten zu verfolgen. Man muß lediglich den Körper in die Nähe eines anderen Librationspunktes bringen. Folgendes Bild zeigt den Satelliten im Librationspunkt L_5 .

- Erde

- Mond



Die Position des Satelliten wurde folgendermaßen berechnet:

$$\text{Positionsvektor: } \vec{x}_{\text{Satellit}} = \begin{pmatrix} \sqrt{0,75} \\ 0,5 - c \end{pmatrix}$$

$$\text{Geschwindigkeitsvektor: } \vec{v}_{\text{Satellit}} = \begin{pmatrix} c - 0,5 \\ \sqrt{0,75} \end{pmatrix}$$

Hierbei wurde der Satellit noch etwas vom Librationspunkt entfernt. Obiges Bild entspricht ungefähr dem Verlauf einer Trojanerbahn. Wegen dieser schleifenartigen Bewegungen um den Librationspunkt spricht man deshalb auch vom *Tanz der Trojaner*.

Literatur

- [1] Gondolatsch, F., *Astronomie I – Die Sonne und ihre Planeten*, Stuttgart, Klett-Verlag, 1989
- [2] Henseling, R., *Astronomie für alle – Sonne und Sternall, 2. Halbband*, Stuttgart, Francksche Verlagshandlung, 1929
- [3] Hempe, K., *Sterne im Computer*, Köln, Verlagsgesellschaft Rudolf Müller, 1986
- [4] Herrmann, J., *Astronomie, eine moderne Sternkunde*, Gütersloh, C. Bertelsmann Verlag, 1960
- [5] *Handbuch Weltall*, Mannheim, Bibliographisches Institut, 1964
- [6] Kippenhahn, R., *Unheimliche Welten*, München, Deutscher Taschenbuch Verlag, 1990
- [7] Lermer, R., *Grundkurs Astronomie*, München, Bayerischer Schulbuch Verlag, 1989

Programmlisting

```
PROGRAM Troja;

{$path "ram:include/","pascal:include/";
incl "intuition.lib","Graphics.lib" }

CONST
    Breite=640; Höhe=512;

TYPE
    Vektor2 = Record
        x, y: real;
    END;

VAR
    NeuWindow:NewWindow;
    MyWindow:^Window;
    rport:^RastPort;
    flag:boolean;
    MyScreen: ^Screen;
    Msg,Upt: ptr;

    mM, mE: Real;
    Mond, Erde: Array[0..360] of Vektor2;
    si, co: Array[0..360] of real;
    Satellit, vSatellit: Vektor2;
    c,t, h: real;
    Nr,i,steps: INTEGER;
    Mitte: Vektor2;
    Einheit: Real;

PROCEDURE GrafikEin; {Zum Öffnen des Fensters}
BEGIN
    OpenLib(IntBase,'intuition.library',0);
    OpenLib(GfxBase,'graphics.library' ,0);
    MyScreen:=Open_Screen(0,0,Breite,Höhe,1,1,2,
        GENLOCK_VIDEO+hires+lance,'Troia.p');
    NeuWindow:=NewWindow(0,0,Breite,Höhe,0,1,
        VANILLAKEY,BORDERLESS+ACTIVATE,
        Nil,Nil,Nil,MyScreen,Nil,100,80,Breite,Höhe,
        CUSTOMSCREEN);
    MyWindow:=OpenWindow(^Neuwindow);
```

```
IF MyWindow=Nil THEN Error('Fenster kann
nicht geöffnet werden!');
SetRGB4(^MyScreen^.ViewPort,0,0,0,0);
SetRGB4(^MyScreen^.ViewPort,1,15,15,15);
SetRGB4(^MyScreen^.ViewPort,2,15,15,15);
rport:=MyWindow^.RPort;
Upt:=MyWindow^.Userport;
END; {GrafikEin}
```

```
PROCEDURE Taste;
BEGIN
    Msg:=Wait_Port(Upt);
    Msg:=Get_Msg(Upt);
    Reply_Msg(Msg);
END; {Taste}
```

```
PROCEDURE Inc360 (VAR i: INTEGER);
BEGIN
    i:=i+1;
    IF i=360 THEN
        i:=0;
    END; {Inc360}
```

```
PROCEDURE Eingabe;
BEGIN
    writeln("Programm : TROIA");
    writeln;

    REPEAT
        write("Massenverhältnisses c?");
        readln(c);
    UNTIL (c<1) and (c>0);

    writeln;
    REPEAT
        write("Wieviele Schritte ?");
        readln(steps);
    UNTIL (steps>0) and (steps <10001);
    writeln;
END; {Eingabe}
```

```
PROCEDURE initSiCo;
```

```
VAR
  i: INTEGER;
  arc: Real;
BEGIN
  arc:= 3.14159265359 /180.0;
  FOR i:=0 TO 90 DO
    BEGIN
      si[i]:= sin(arc*i);
      si[180 - i] := si[i];
      si[180 + i] := -si[i];
      si[360 - i] := -si[i];

      co[i]:= cos(arc*i);
      co[360 - i] := co[i];
      co[180 + i] := -co[i];
      co[180 - i] := -co[i];
    END;
  END; {initSiCo}

Function f0(rp:real):real;
BEGIN
  f0:=(1-c)/sqr(c+rp)-c/sqr(1-c-rp)-rp;
END; {f0}

Function f1(rp:real):real;
BEGIN
  f1:=2*(c-1)/sqr(c+rp)/(c+rp)-
      2*c/sqr(1-c-rp)/(1-c-rp)-1;
END; {f1}

PROCEDURE Anfangswerte;
VAR
  xa,x,y,c,t,m:real;
  i:INTEGER;

BEGIN

  x:=0

  REPEAT {Newtonsches Iterationsverfahren zur
    Berechnung der Nullstelle}
    xa:=x;m:=f1(x);y:=f0(x);
    t:=y-m*x;
```

```
x:=-t/m;
UNTIL abs(f0(x))<0.001;

writeln("Nullstelle: f(",x,") = ",f0(x));

Satellit.x:=0;
Satellit.y:=x;
vSatellit.x:=-x;
vSatellit.y:=0;

Mitte.x:=Breite/2.5;
Mitte.y:=Höhe/2;
Einheit:=Breite/2;
END;

PROCEDURE Init;
VAR
  i: INTEGER;
BEGIN
  mM:=c;
  mE:= 1.0 - mM;

  FOR i:= 0 TO 360 DO
    BEGIN
      Mond[i].x := -si[i] * mE;
      Mond[i].y := co[i] * mE;
      Erde[i].x := si[i] * mM;
      Erde[i].y := -co[i] * mM;
    END;

    t := 0.0;
    Nr := 0;
    h := 3.1415926 /90.0;

    Anfangswerte;

  END; {init}

PROCEDURE Plot (s: Vektor2);
BEGIN {Plot}
  Draw(rport, Round((s.y * co[Nr] - s.x *
```

```

        si[Nr] ) * Einheit + Mitte.x), Round((s.x
        * co[Nr] + s.y * si[Nr]) *
        Einheit + Mitte.y));
END; {Plot}

```

```

PROCEDURE PlotStart (s: Vektor2);
BEGIN {PlotStart}
    Move(rport, Round((s.y * co[Nr] - s.x *
        si[Nr]) * Einheit + Mitte.x),
        Round((s.x * co[Nr] + s.y * si[Nr]) *
        Einheit + Mitte.y));
END; {PlotStart}

```

```

PROCEDURE PlotJuSo;
VAR
    x0, y0: INTEGER;
BEGIN
    x0 := Round(Erde[0].y * Einheit + Mitte.x);
    y0 := Round(Erde[0].x * Einheit + Mitte.y);
    RectFill(rport, x0-1, y0-1, x0+2, y0+1);
    Move(rport, x0+5, y0+2);
    flag := Text(rport, 'Erde', 4);

    x0 := Round(Mond[0].y * Einheit + Mitte.x);
    y0 := Round(Mond[0].x * Einheit + Mitte.y);
    RectFill(rport, x0-1, y0-1, x0+2, y0+1);
    Move(rport, x0+5, y0+2);
    flag := Text(rport, 'Mond', 4);
END; {PlotJuSo}

```

```

PROCEDURE f (Satellit: Vektor2; VAR a: Vektor2);
VAR
    rM2, rE2, rM3, rE3: Real;
    rM, rE: Vektor2;
BEGIN
    rM.x := Mond[Nr].x - Satellit.x;
    rM.y := Mond[Nr].y - Satellit.y;
    rE.x := Erde[Nr].x - Satellit.x;
    rE.y := Erde[Nr].y - Satellit.y;
    rM2 := sqr(rM.x) + sqr(rM.y);
    rE2 := sqr(rE.x) + sqr(rE.y);

```



```

    rM3 := rM2 * sqrt(rM2);
    rE3 := rE2 * sqrt(rE2);
    a.x := mE * rE.x / rE3 + mM * rM.x / rM3;
    a.y := mE * rE.y / rE3 + mM * rM.y / rM3;
END;

```

```

PROCEDURE Nyström2 (VAR s,v:Vektor2;,t,h:real);
VAR

```

```

    h2: Real;
    k1,k2,k4,x1,a1,a2,a3: Vektor2;
BEGIN
    h2 := h * 0.5;
    f(s,a1);

    k1.x := a1.x * h2;
    k1.y := a1.y * h2;
    t := t + h2;
    x1.x := s.x + (v.x + k1.x * 0.5) * h2;
    x1.y := s.y + (v.y + k1.y * 0.5) * h2;
    Inc360(Nr);
    f(x1,a2);

    k2.x := a2.x * h2;
    k2.y := a2.y * h2;
    t := t + h2;
    x1.x := s.x + (v.x + k2.x) * h;
    x1.y := s.y + (v.y + k2.y) * h;
    Inc360(Nr);
    f(x1,a3);

    k4.x := a3.x * h2;
    k4.y := a3.y * h2;

    s.x := s.x + (v.x + (k1.x + k2.x * 2)/3)* h;
    s.y := s.y + (v.y + (k1.y + k2.y * 2)/3)* h;
    v.x := v.x + (k1.x + k4.x + k2.x * 4)/3;
    v.y := v.y + (k1.y + k4.y + k2.y * 4)/3;
END;

```

```

BEGIN

```

```
Eingabe;
initSiCo;
init;
GrafikEin;
PlotJuSo;
PlotStart(Satellit);
FOR i:=0 TO steps DO
BEGIN
    Nyström2(Satellit, vSatellit, t,h);
    Plot (Satellit);
END;
Taste;
END.
```

4 Erklärung

Ich erkläre hiermit, daß ich die Facharbeit ohne fremde Hilfe angefertigt und nur die im Literaturverzeichnis angeführten Quellen und Hilfsmittel benützt habe.

Höttingen, den 16. Januar 1991 _____